
Subject: Re: spherical harmonic representation
Posted by [Kenneth P. Bowman](#) on Tue, 21 Jul 2009 14:16:04 GMT
[View Forum Message](#) <> [Reply to Message](#)

In article

<44077862-b715-4e61-a59c-0d4e390c4e6a@r2g2000yqm.googlegroups.com>,
maurya <ramaurya98@gmail.com> wrote:

> Hi everybody,
>
> I want to display spherical harmonics on a sphere, as given on web
> page:
>
>
> But, it is not giving results as expected (as given on the above web
> page).
>
> Can anyone help me.
>
> Thanks

There are two immediate errors that I see. Theta is the co-latitude, not the latitude, and should range from 0 to 180, not -90 to 90. If you want to plot the whole sphere, the longitude angle phi should range from 0 to 360 (or -180 to 180), not -90 to 90.

When I fix these things there is still something wrong. The longitudinal variation should be a single wave-5 sinusoid, but it is not. Either I am still missing something, or there is a bug in SPHER_HARM.

(I changed a few details to make it easier for me to understand.)

Ken Bowman

pro sph_plot

```
l=5      ;harmonic degree
m=2      ;azimuthal order |m| <= l
np = 101 ;No. of inclination angle(theta) and azimuthal angle(phi)

x = 360.0D0*DINDGEN(np)/(np-1)      ;phi = 0, 360
y = 180.0D0*DINDGEN(np)/(np-1)      ;theta = 0, 180
xx = REBIN(x, np, np)
yy = REBIN(REFORM(y, 1, np), np, np)

ff = real_part(spher_harm(xx*!dtr, yy*!dtr, l, m, /doub))
;ff = IMAGINARY(spher_harm(xx*!dtr, yy*!dtr, l, m, /doub))
```

```
window, 0, xsize = 512, ysize = 512, title = 'Spherical Harmonic'
```

```
MAP_SET, /CYLINDRICAL
```

```
contour, ff, x, y-90.0, /ove, nlevels=10
```

```
isurface, ff, x, y-90.0
```

```
end
```

Subject: Re: spherical harmonic representation

Posted by [Kenneth P. Bowman](#) on Tue, 21 Jul 2009 14:55:33 GMT

[View Forum Message](#) <> [Reply to Message](#)

In article <k-bowman-C664B6.09160321072009@news.tamu.edu>,
"Kenneth P. Bowman" <k-bowman@null.edu> wrote:

Had the theta and phi arguments switched. This works.

Ken

```
pro sph_plot
```

```
l=5      ;harmonic degree
```

```
m=2      ;azimuthal order |m| <= l
```

```
np = 101  ;No. of inclination angle(theta) and azimuthal angle(phi)
```

```
x = 360.0D0*DINDGEN(np)/(np-1)      ;phi = 0, 360
```

```
y = 180.0D0*DINDGEN(np)/(np-1)      ;theta = 0, 180
```

```
xx = REBIN(x, np, np)
```

```
yy = REBIN(REFORM(y, 1, np), np, np)
```

```
ff = real_part(spher_harm(yy*!dtor, xx*!dtor, l, m, /doub))
```

```
window, 0, xsize = 512, ysize = 512, title = 'Spherical Harmonic'
```

```
MAP_SET, /CYLINDRICAL
```

```
contour, ff, x, y-90.0, /ove, nlevels=10
```

```
isurface, ff, x, y-90.0
```

```
end
```
